Actalk: A Framework for Object-Oriented Concurrent Programming - Design and Experience

Jean-Pierre Briot
Laboratoire d'Informatique de Paris 6 (LIP6)
Université Paris 6 - CNRS
Case 169, 4 place Jussieu, 75252 Paris Cedex 05, France
E-mail: Jean-Pierre.Briot@lip6.fr

ABSTRACT. This paper describes a framework for modeling, classifying, *and* specializing *various object-oriented concurrent programming (OOCP) languages constructs and schemes. This framework, named* Actalk, *is implemented and integrated within the Smalltalk-80 programming environment. The architecture of the framework is based on the notions of* components, parameterization *and* inheritance. *The kernel of the architecture is decomposed into a set of kernel* component classes *and associated* parameter methods, *to be redefined in various subclasses. Various* extensions, *defined as* subclasses *of the kernel classes, simulate various OOCP languages and constructs. They implement various models of* activity *(serialized, concurrent, with implicit or explicit message acceptance...),* communication *(synchronous, eager reply, time stamped...), and* synchronization schemes *(guards, abstract states transitions, synchronization counters...). At first, our framework helps in classifying and experimenting with various synchronization and communication models for a given program, by changing and specializing various models/components. Secondly, it helps in reusing them and building-up on previous expertise in order to derivate enhanced or novel constructs or schemes. To illustrate the possibilities in this paper, we progressively develop various synchronization schemes by successive* refinements *and* combinations, *starting from two foundational synchronization schemes (namely* enabled sets *and* guards*).*

Keywords: concurrency, programming, languages, communication, synchronization, schemes, framework, platform, components, parameterization, refinement, combination, implementation, Smalltalk-80.

1 Introduction

Object-oriented concurrent programming (OOCP in short [YT87]) is an important methodology for development of parallel, distributed and open applications [BGY96]. It is a generalization of object-oriented programming taking

into account multiple activities. OOCP provides both: (1) a methodology for the *decomposition* of complex programs which may expressed more naturally as a coordination of multiple interacting entities, and (2) a means for exploiting perspectives of speed-up as offered by new parallel and distributed computing architectures.

As a consequence of the success of this approach, various programming models, constructs and mechanisms have been proposed (e.g., see in [YT 87], [AWY 93], [BGY 96]). They reflect a wide variation of concerns and domains. This variety of models, and the difficulty to compare them by abstracting from various associated terminologies, syntax and implementational foundations, led us to design some small comprehensive classification testbed in 1988 [BRI 89]. In a second step, this testbed was later expanded in 1994 into a full framework [BRI 96]. After several years of successive developments, experiments, usage by various communities for different application domains and concerns, we would like to summarize the results and lessons learned. This paper aims at providing such description and assessments.

1.1 Outline

This paper is organized as follows. In Sect. 2, we first discuss the motivations for the Actalk platform. In Sect. 3, we then discuss the precise objectives of the platform and the main design decisions we made. Section 4 describes the architecture and more specifically the way the kernel is decomposed and parameterized. In Sect. 5, we progressively develop various schemes by successive *refinements* and *combinations*, starting from two foundational synchronization schemes (namely *enabled sets* and *guards*). Section 6 quickly summarizes and evaluates our approach and results, and then compares them to other relevant works before concluding this paper.

2 Context and Motivations

2.1 Object-Oriented Concurrent Programming

Object-oriented (concurrent) programming (OOCP) is based on a few simple and generic concepts: (active) objects and message passing. Concepts are strong enough to help at structuring and encapsulating computation modules, and generic enough to encompass various software and hardware architectures.

OOCP *generalizes* standard (sequential) object-oriented programming. The basic idea is to integrate object notions with the main abstractions of concurrent programming.[1] It identifies the notion of object with the notion of activity,

[1] The objective of this integrative approach is to provide a simple and unified conceptual model to the programmer. Note that this is not the only way to relate object concepts with concurrency, parallelism, and distribution concerns. [BGL 98] considers three main approaches (applicative, integrative, reflective) and studies their specificities, differences as well as their complementary aspects and levels.

leading to the notion of an *active object*. It also identifies message passing between objects with the synchronization of the sender (client) onto the receiver (server). This eases the issue of synchronization of concurrent programs, as synchronization of requests becomes transparent to the client.

2.2 Various Dimensions

Based on these general concepts, various alternative programming models, constructs and mechanisms have been proposed in various programming languages and systems [YT 87, AWY 93, BGY 96]). These various systems reflect a wide variation of concerns and domains. Possible variations on programming language design may be grouped into various aspects, or dimensions, such as:

- *communication*: is communication between objects synchronous or asynchronous ?, with a single reply or multiple replies ? are there priority schemes ?, etc.
- *activity*: is acceptance of messages implicit or explicit ?, is there intra-object concurrency (i.e., may an object compute several requests simultaneously) ?, etc.
- *synchronization/coordination*: how does an active object control message selection and activation ? depending on what information ? (its state, current invocations, required services, etc.) and along which scheme ?[2]

2.3 Motivations

This variety of models, and the difficulty to compare them by abstracting from various associated terminologies, syntax and implementational foundations, led us to design a comprehensive and unified modeling testbed.

This testbed has been developed in, and integrated within, the Smalltalk-80 programming environment. It is named *Actalk*, which stands for *act*ive objects (or *act*ors) in Small*talk*. It is a framework and is based on the notions of *components*, *parameterization*, and *inheritance*. Various component classes correspond roughly to various dimensions, as defined above.

Motivations are both:

- to help at the *analysis* and *classification* of different aspects of various formalisms (computation models, programming langagues, constructs, schemes), of which different syntax and implementation frameworks may hide the semantics.

[2] Finding schemes which allow easily reusable specifications is actually non trivial. This problem has been named the *"inheritance anomaly phenomenon"* by Satoshi Matsuoka [MY 93]. Although this paper is not about proposals for solving the anomaly, we will show in Sect. 5 that our platform already includes most of schemes which have been proposed, and moreover helps at adapting them.

- to help at the *selection* and *comparison* with various computation models, by associating various components (e.g., various synchronization schemes, taken from the existing library of components) to a given program.
- to ease at the *design* of new formalisms, by *refinement* and *combination* of existing components. (An example of successive refinements and combinations will be developed in Sect. 5).
- to provide an environment for active experiment, with a rich *programming environment* including visualization and debugging tools.

3 Objectives and Design Decisions

3.1 Objectives

To design a system for integrating various OOCP models and languages in a single environment, we had the following objectives in mind:

- *uniformity.* Various object-oriented concurrent programming computation models, languages constructs, and schemes are grouped and classified into one *single* framework.
- *minimality.* Actalk is based on a *kernel* which provides the basic and default semantics of OOCP languages.
- *modularity.* The kernel is decomposed into a set of kernel *component classes* and their associated *parameter methods.*
- *extensibility and expressivity.* Various extensions, defined as *subclasses* of the kernel classes and which specialize some of the parameter methods, simulate various OOCP languages and constructs.
- *simplicity.* We are not concerned in this platform with considerations for optimization or formal specifications, as they may complexify the implementation.
- *integration.* Actalk is integrated within the Smalltalk-80 programming environment. Thus the user may combine various object-oriented concurrent programming styles with standard object-oriented programming, as provided by Smalltalk-80, to reuse standard programs and to express various program granularities.
- *experiment.* Actalk includes a library of various component classes describing various language constructs, models of activity, synchronization, and communication, and associated example programs. Standard Smalltalk-80 programming environment provides various development tools to develop schemes and programs.[3]

[3] In fact some of the Smalltalk-80 standard tools, such as the MVC user interface framework, the scheduler, and the debugger, need some adjustements for dealing properly with active objects. A past project has developed prototype extensions of Smalltalk-80 standard tools, such as an extended MVC model for active objects, and a generic scheduler and its associated visualization tools [LES 92].

3.2 Choosing Smalltalk

Smalltalk-80 was chosen as a foundation for its high-level, flexible and rich environment. Smalltalk-80 provides all entities needed to build active objects (objects, classes), and to manage their activities (processes, message perform), their communication (messages, queues), and their synchronization (semaphores, shared queues) [BG 96].[4]

4 Architecture of the Actalk Framework

4.1 General Architecture

The architecture of Actalk includes a *kernel* which models basic OOCP semantics (that is, serialized active objects which communicate by asynchronous unidirectional message passing). The kernel is composed of a set of *kernel component classes.* Each *component* describes a different aspect of an active object, that is: behavior, activity/synchronization, and communication. Each component class is parameterized, that is some of its functionalities (methods) are specifically intended to be specialized in subclasses in order to model alternative language designs. Therefore, these (virtual) methods are named *parameter methods.*

Note that the granularity and balance of the decomposition of the architecture into components and parameter methods is a very sensitive issue. Finer granularity of decomposition brings greater modularity but at the cost of possible complexity, as well as increasing consistency management and efficiency problems. This issue will be discussed in Sect. 6.3, when comparing our platform to a few other similar systems.

4.2 Component Classes of the Actalk Kernel

The three main kernel component classes are:

- Class `ActiveObject` describes the *behavior* of the active object, that is the inner standard object which ultimately computes the messages. Application classes of active objects are defined as subclasses of class `ActiveObject`. The language designer may also implement specific programming language constructs in some abstract subclasses. (For instance subclass `ActorObject` implements the Actor model of computation construct `replace` for specifying the replacement behavior of an actor [AGH 86]. Class `Abcl1Object` implements the ABCL/1 language

[4] One partial meta-representation of Actalk, where Actalk active objects may be described by a set of active objects, has been developed by Sylvain Giroux and named ReActalk [GS 91]. ReActalk increases the flexibility of Actalk through reflection, at the cost of a little more complexity and little less efficiency.

construct **wait-for**, to explicit wait for some message pattern [YT87, pages 55–89].)

- Class **Activity** describes the *internal activity* of the active object. This class provides autonomy[5] for the active object. It also defines the way method invocations are selected, scheduled and computed. Consequently, its subclasses may describe various synchronization schemes.

- Class **Address** describes the *address* (mailbox) of an active object, that is the identifier of an active object to which messages will be sent. This class defines the way message transmissions will be interpreted. Its subclasses may implement various types of communication.[6]

This decomposition allows the independent modeling of various dimensions of active objects. One can decouple the actual program from a given communication model (e.g., with eager reply), and from a given model of activity (e.g., with intra-object concurrency) and synchronization scheme[7] (e.g., guards). Meanwhile, combination of arbitrary components may lead to inconsistencies. Therefore the Actalk platform includes some simple compatibility specification and verification mechanism to keep some safeguards [BRI94].

Note that the kernel actually provides two more kernel component classes: class **MailBox** which represents the message queue, and class **Invocation** (a subclass of standard class **Message**) which represents a method invocation. They are not considered as prime components because their parameterization is minimal. Meanwhile their functionality may be extended by subclassing them as for the three main component classes. This proves to be useful when modeling complex protocols for message buffering, e.g., with priorities, or for invocation management, e.g., with time stamps (to be described in Sect. 5.5).

4.3 Relations between Components

To define a class of active objects, the programmer should define this class as inheriting from class **ActiveObject** (and not **Object** as usual). After creating an active object behavior as its instance, creation and initialization of the associated components (activity, address and mailbox) take place transparently. Default classes for associated components are expressed by specific parameter

[5] through a (light weight) standard Smalltalk-80 process.

[6] In Actalk, the receiver defines the semantics of message passing (synchronous, asynchronous...). This simple scheme provides modularity and encapsulation, as well as simple and efficient implementation. It may be extended into a more general framework where the type of communication is negotiated between sender and receiver, at the cost of more complexity and less efficiency.

[7] Note that in the architecture, the activity/synchronization component is explicitly and structurally distinct from the behavior/program component. This enforces independence between program code/data and synchronization code/data as advocated in [MCH94]. This also eases comparison of various synchronization schemes on actual examples by just changing the synchronization class while keeping the same behavior/program.



Figure 1: Components of an Actalk active object

methods (e.g., parameter method **addressClass** specifying the associated address component class, as later detailed in Table 1). This follows the *Factory method* design pattern [GHJ+95].

The minimal (default) functionality of an active object, and the relations and interactions between its components, may be summarized by following the main steps of a message, from its reception to its processing (see Fig. 1):

- the address *receives* a message, triggering parameter method **receiveMessage:** of class **Address** which enqueues the message in the mailbox ;

- *independently*, the activity *selects* the next message to be processed (parameter method **nextMessage** of class **Activity**) ;

- and accepts it (parameter method **acceptMessage:** of class **Activity**), ultimately delegating its processing (parameter method **performMessage:** of class **Activity**) to the behavior.

4.4 The Activity Kernel Component Class and its Parameterization

Class **Activity** defines two instance variables: **address**, to reference the associated address component, and **bself** (as for *b*ehavior *self*), to reference the associated behavior component, of the active object. Parameter methods of class **Activity** are summarized in Table 1.

Table 1: **Table 1.** Parameter methods of component class `Activity`

method selector	*parameter*	*default value*	*example of redefinition*
`initialize`	*initialization*	none	initialize synchronization counters (see Sect. 5.3)
`start`	*start the activity (process)*	start the process created by `createProcess`	start a second activity process specific to ABCL/1 express mode messages
`createProcess`	*create the activity process*	create a process computing `body`	provide a handle to the process (useful for termination control)
`body`	*specification of the activity*	serially accept successive messages	accept a single message (e.g., Actors model of activity)
`nextMessage`	*next message to be accepted*	return and remove first message from the message queue	return and remove the first message whose selector is enabled (see Sect. 5.1)
`acceptMessage:`	*accept and compute a message*	`performMessage:`	spawn an inner thread to compute the message (intra-object concurrency class `ConcurrentActivity`)
`performMessage:`	*perform a message*	delegate actual perform to the behavior	also return the value to the implicit reply destination
`addressClass`	*default address component class*	`Address`	`PoolAddress` (associated to class `PoolActivity`)
`invocation-ClassFor:`	*invocation class*	`Message`	`WithSenderInvocation` (includes the sender)
`eventReceive:/ Accept:/ Complete:`	*message reception/ acceptance/ completion generic events*	none	after completion, compute the next enabled set of selectors (see Sect. 5.1)

4.5 Generic Event Methods

Another important characteristic of the Actalk architecture is the existence of *generic event methods*. These parameter methods are associated to the three following events: *reception/acceptance/completion* of an invocation. They are respectively named: `eventReceive:/Accept:/Complete:`. (They take the current invocation as their argument.)

The user may redefine them to attach actions to a given class of active objects, e.g.: for tracing activities, stepping computation, controlling global scheduling of activities... Generic event methods may also be used by the designer for modeling language specificities (e.g., for computing post actions as with the POOL language [YT 87, pages 199–220]). Last, they are also very useful for managing intra-object synchronization. Subclass `SynchroConcurrent-Activity` specializes them in order to ensure their atomicity (mutual exclusion), as we will see in Sect. 5.3.

4.6 Layered Architecture

In complement of its component-based architecture (Sect. 4.2), and its event-based architecture (Sect. 4.5), the kernel is also designed along some layered architecture.[8] The three successive layers[9] help at separating *basic properties*, *added facilities*, and *(public) entry points* for the user.

The first layer defines *minimal* versions of the three main component classes of the kernel, including basic methods and parameter methods. The second layer adds more facilities, notably the event mechanism. The third layer defines the three kernel component classes `ActiveObject`, `Activity` and `Address`, to be subclassed by the programmer. They also include some user-intended functionalities (tracing, compatibility constraints...).

4.7 Libraries

Various *extensions* of the Actalk kernel have been defined as subclasses of one or more of the kernel component classes. These various component subclasses simulate various:

- *language models and constructs*: e.g., Actors concept of behavior replacement [AGH 86], ABCL/1 explicit wait for a message pattern [YT 87, pages 55–89], POOL concept of body and post actions construct [YT 87, pages 199–220]... ;
- *communication models*: e.g., ABCL/1 three types (synchronous, asynchronous, future) and two modes (normal, express) of message transmission, implicit reply mechanism (for a good integration within underlying standard Smalltalk-80 method execution model)... ;

[8] This shows an example of combining three different software architectural styles [SG 96] within a single system.

[9] They were introduced by Loïc Lescaudron [LES 92].

- and *synchronization schemes*: the focus of this paper. Part of the corresponding taxonomy is detailed in Sect. 5. Other schemes include: explicit message acceptance (as in POOL and Eiffel//), method suspension (as in Portable ConcurrentSmalltalk), reflective framework (as in OCore)...

4.8 Other Developments and Applications

With its libraries and taxonomies of various OOCP languages characteristics, the Actalk framework provides a comprehensive view of design alternatives and mechanisms. Actalk has been used as a tool for teaching and experiments by various people, inside Paris 6 University, but also in other places, notably in the early 90's during graduate courses led by Jean Bézivin at University of Nantes where students projects developed many experiments.

The kernel of Actalk has also been used as component or foundation for several projects and domains, such as simulation of software engineering process models [KG 94], distributed constraint satisfaction algorithms (e.g., [GHÉ 94]), the study of exception mechanisms for object-oriented concurrent programming, etc. One notable field of application has been for construction by various teams of various prototype multi-agent systems, themselves applied to various domains (eco-systems simulation, distributed problem solving [BFS 91], natural language processing [NH 96], knowledge acquisition [BH 94]...). A recent example multi-agent platform, and the way it is built up from Actalk active objects, is described in [GB 98].

5 A Developed Example: Modeling Synchronization Schemes

We will now model and implement various synchronization schemes and show the expressiveness of our platform. As noted in the introduction, *enabled sets* and *guards* are the two major synchronization schemes foundations. We will use successive refinements and combinations of these two schemes to produce increasingly expressive (and complex) synchronization schemes. Our point is actually in showing how we may easily enhance and customize various synchronization schemes along various requirements, while building-up on previous expertise. Figure 2 summarizes the hierarchy of synchronization schemes/classes which will be developed in this section.

Note that we sometimes combine two synchronization schemes into a new one (e.g., class `CountersActivity` described in Sect. 5.3), thus inheriting from *more* than one class. As there is currently no multiple inheritance mechanism in Smalltalk-80, we unfortunately must choose a *single* superclass (*solid arrow* in Fig. 2) and recopy variables and methods from the other one(s) (*dashed arrow*). These copied methods won't be shown in the definitions given in the paper.



Figure 2: Hierarchy of synchronization schemes/classes (sample)

5.1 A First Basic Scheme: Enabled Sets/Abstract States

The first synchronization scheme that we implement is the *enabled sets* synchronization scheme. This scheme is very good at expressing *services availability* constraints, i.e. how an active object is willing to accept certain method invocations. In many cases, parameters are not significant in the decision. One may then introduce a further level of abstraction in enabling (or disabling) method patterns (i.e. selectors) grouped into *enabled methods sets*.

Abstract states (also named *behavioral abstractions*) are abstractions/names to which enabled sets will be assigned. (In the canonical example of the bounded buffer, `put:` requests should be disabled while the buffer is full. This will be expressed with an abstract state `full` having only {`get`} as its corresponding enabled set.) After selecting and computing the current method invocation, a state transition function will compute the next abstract state (leading to the next enabled set). Enabled sets are computed through set operations such as union (`+`), intersection (`&`), and difference (`-`).

Class `EnabledSetsActivity` implements this scheme in the following way. It defines an instance variable `enabledSelectors` to hold the current enabled set. An enabled set of selectors is implemented as a Smalltalk array (e.g., `#(get put:)`).[10] An abstract state is represented by a method returning an enabled set. Implementation only redefines three of class `Activity` parameter methods: `initialize`, `nextMessage`, and *completion* generic event method `eventComplete:`.

```
Activity subclass: #EnabledSetsActivity
   instanceVariableNames: 'enabledSelectors '

"Parameter method for initialization of the activity."
initialize
   super initialize.
   "Initialize the initial set of enabled selectors."
```

[10] `#` is the quotation character in Smalltalk, both for symbols (e.g., `#empty`) and for arrays (e.g., enabled set `#(put:)`).

```
    enabledSelectors := self perform: self initialAbstractState

"Return first message from the mailbox belonging to current enabled set."
nextMessage
    ^self mailBox firstMessageWithCondition:
         [:message | enabledSelectors includes: message selector]

"Generic event method associated to completion of the invocation (computation)."
eventComplete: aMessage
    super eventComplete: aMessage.
    "Abstract state transition: computing next set of enabled selectors."
    enabledSelectors :=
       self perform: (self nextAbstractStateAfter: aMessage selector)
```

Note that method definitions reference (call) two undefined methods: **initialAbstractState** (to specify the initial abstract state) and **nextAbstractStateAfter:** (to compute the next abstract state). These two *virtual* methods are not defined at this abstract level, but must be in a specific (concrete) synchronization subclass (e.g., in next example). Note that the code above is a concise but *complete* implementation.

Example: Synchronization of a Bounded Buffer (1)

The enabled sets/abstract states synchronization scheme is very good at expressing services availability constraints based on the state of the (active) object. In the canonical example of the bounded buffer, there will be three abstract states (note that abstract state **partial** may be defined as the union of **empty** and **full**):

abstract state	*enabled set*
empty	#(put:)
partial	#(get put:)
full	#(get)

The synchronization component of the bounded buffer is specified in class **BufferEnabledSetsActivity**, defined as a subclass of class **EnabledSetsActivity**:

```
EnabledSetsActivity subclass: #BufferEnabledSetsActivity
    instanceVariableNames: ''

"Abstract states."
empty
    ^#(put:)

full
    ^#(get)

partial
"Defined as the union (+) of the abstract states empty and full."
    ^(self empty) + (self full)
```



```
"The initial abstract state when creating an instance."
initialAbstractState
    ^#empty

"Abstract state transition: computing the next abstract state."
nextAbstractStateAfter: selector
    ^bself isEmpty
        ifTrue: [#empty]
        ifFalse: [bself isFull
                    ifTrue:  [#full]
                    ifFalse: [#partial]]
```

Note that the implementation of the bounded buffer behavior/program (trivial and here out of scope) is not shown in this paper. We assume that the behavior implements methods **put:** and **get**. We also assume that it implements the two predicate methods **isEmpty** and **isFull**, and the accessor method **maxSize** (to consult the maximum size). As opposed to methods **put:** and **get**, these latter methods are not declared in the external interface of the active object (i.e. they won't be enabled). They are for internal use only, so that the activity/synchronization component may (namely, method **nextAbstractStateAfter:**) consult the state of the buffer behavior (referenced by instance variable **bself**, see Sect. 4.4).

5.2 A Second Basic Scheme: Guards

The main alternative to enabled sets are *guards*. Intuitively, a method invocation will be blocked until the guard (boolean activation condition) associated to the method evaluates to "true". Class **GuardsActivity** implements a simple and naive mechanism for guards.[11] Parameter method **nextMessage** is redefined in order to look for the first candidate message (method **isCandidateMessage:**) whose corresponding guard evaluates to **true**. It keeps fetching the next message and re-enqueueing it into the mailbox (method **internalReceiveMessage:**) until it finds a candidate. Finally, we represent a guard associated to a method as another method (whose name is prefixed by symbol **guardOF**). The *complete* implementation is as follows:

```
Activity subclass: #GuardsActivity
    instanceVariableNames: ''

"Return the first candidate message, otherwise re-enqueue it"
    (method internalReceiveMessage:)."
nextMessage
    | message |
    [message := super nextMessage.
     self isCandidateMessage: message] whileFalse:
```

[11] Note that we also provide refinements of this naive initial implementation scheme without resending messages. They use indexing messages within the mailbox and furthermore implement some optimized, but safe, reevaluation semantics for guards. They are implemented in subclasses of class **GuardsActivity** but won't be described in this paper.

```
        [address internalReceiveMessage: message].
    ^message

"A message is candidate for acceptance if the associated guard evaluates to true."
isCandidateMessage: aMessage
    ^self evaluateGuardForMessage: aMessage

"Evaluate the associated condition/guard method with the current arguments."
evaluateGuardForMessage: aMessage
    ^self perform: (self guardOfSelector: aMessage selector)
          withArguments: aMessage arguments

"Return the selector of the associated guard: selector prefixed by guardOF."
guardOfSelector: selector
    ^('guardOF' , selector) asSymbol
```

Example: Synchronization of a Bounded Buffer (2)

Services availability constraints on the bounded buffer are easily expressed as guards referring to the state of the buffer:

```
GuardsActivity subclass: #BufferGuardsActivity
    instanceVariableNames: ''

"Guards: boolean activation conditions associated to methods."
guardOFget
    ^bself isEmpty not

guardOFput: item
    ^bself isFull not
```

Note that guards have the advantage over enabled sets that they may decide on the acceptance of a particular message based on some of its parameters. But efficient implementations of guards is more difficult to achieve.

5.3 A First Refinement: Guards Extended with Synchronization Counters

With the addition of some fine grain mechanism observing status of current invocations, guard notations may easily express intra-object synchronization, that is control over multiple method invocations within a single active object. *Synchronization counters* [RV 77] are often chosen as such a general and expressive mechanism. Main counters record the number of invocations *received*, *accepted*, and *completed* for a given method selector. Actually, synchronization counters are a direct consequence of (Actalk) generic events.

We also need a specialized activity class providing intra-object concurrency, as implemented by class `ConcurrentActivity`. Its subclass `SynchroConcurrentActivity` redefines generic event methods in order to ensure their atomicity (mutual exclusion), by introducing some mutual exclusion semaphore (instance variable `mutexSemaphore`).

Now, we may implement synchronization counters (class `CountersActivity`) as a subclass of both class `SynchroConcurrentActivity` (in order to inherit concurrency and atomic events) *and* class `GuardsActivity` (to inherit guards). The only difference with class `GuardsActivity` (Sect. 5.2) is the redefinition of method `isCandidateMessage:` in order to ensure atomicity of a successful guard evaluation with the *acceptance* event (method `eventAccept:`). Class `CountersActivity` also implements dictionaries to record and consult synchronization counters data.

```
SynchroConcurrentActivity subclass: #CountersActivity
    instanceVariableNames: 'receivedCounterDictionary
          acceptedCounterDictionary completedCounterDictionary '

initialize
    super initialize.
    "Create a dictionary per family of counters, indexed by each selector."
    self makeSynchroCounterDictionariesOnSelectors:
          bself class allScriptSelectors

"A message is candidate for acceptance if its associated guard evaluates to true."
isCandidateMessage: aMessage
    "Note that successful guard evaluation AND acceptance event must be atomic."
    ^mutexSemaphore critical:
        [(self evaluateGuardForMessage: aMessage)
            ifTrue: [self eventAccept: aMessage.
                     true]
            ifFalse: [false]]

"Reception event method increments associated synchronization counter."
"(Same for methods eventAccept: and eventComplete:.)"
eventReceive: aMessage
    super eventReceive: aMessage.
    receivedCounterDictionary at: aMessage selector
       put: (receivedCounterDictionary at: aMessage selector) + 1

"Consultation of the reception synchronization counter."
"(Same for methods accepted: and completed:.)"
received: selector
    "Number of received invocations of message selector."
    ^receivedCounterDictionary at: selector

"Simulation of other useful synchronization counters."
current: selector
    "Number of current (accepted but not completed yet) invocations of selector."
    ^(self accepted: selector) - (self completed: selector)

pending: selector
    "Number of pending (received but not accepted yet) invocations of selector."
    ^(self received: selector) - (self accepted: selector)
```

Example: Synchronization of a (Concurrent) Bounded Buffer (3)

Suppose that we now free the internal concurrency of a bounded buffer.[12] We then must ensure its internal consistency, by forbidding concurrent processing of several `put:` invocations (and as well for `get` invocations). On the other hand, simultaneous processing of one `put:` *and* one `get` is allowed, as they access distinct memory sectors. Note that the number of items of the buffer is computed as the difference between completed `put:` and completed `get`, thus only relying on synchronization data.

```
CountersActivity subclass: #BufferCountersActivity
   instanceVariableNames: ''

"Guards."
guardOFget
   "Only one get at once AND the buffer is not empty."
   ^(self current: #get) = 0
      and: [(self completed: #put:) - (self completed: #get) > 0]

guardOFput: item
   "Only one put: at once AND the buffer is not full."
    ^(self current: #put:) = 0
      and: [(self completed: #put:) - (self completed: #get)
                                          < bself maxSize]
```

5.4 A Combination: Enabled Sets with Guards/Synchronization Counters

Note that enabled sets (Sect. 5.1) are specific to services availability constraints, whereas guards with synchronization counters (Sect. 5.3) add intra-object synchronization constraints. It is therefore natural to try to combine them into a single scheme, for a clear separation of services availability and intra-object concurrency. Such a mixed scheme has initially been introduced in the Dooji language by Laurent Thomas [THO 94].

Its specification is easily achieved in Actalk by combining class `EnabledSetsActivity` with class `CountersActivity` (the actual superclass) into subclass `EnabledSetsCountersActivity`. The key aspect is the *atomic* combination of the two (enabled sets *and* guards) synchronization conditions (in method `evaluateGuardForMessage:`). We also combine "by hand" initialization (`initialize`) and *completion* event (`eventComplete:`) parameter methods, as shown below:

```
CountersActivity subclass: #EnabledSetsCountersActivity
   instanceVariableNames: 'enabledSelectors '

evaluateGuardForMessage: aMessage
```

[12] Please keep in mind that the example of the bounded buffer is pedagogical and simple, but should *not* be considered as some significant example/granularity of object on which to introduce intra-object concurrency.


```
   "Check both conditions."
   "(Atomicity is ensured by the call from method isCandidateMessage:,"
   "    see in Sect. 5.3.)"
   ^(enabledSelectors includes: aMessage selector)
      and: [super evaluateGuardForMessage: aMessage]

"By-hand combinations of methods from the two superclasses."
initialize
   super initialize.
   enabledSelectors := self perform: self initialAbstractState

eventComplete: aMessage
   super eventComplete: aMessage.
   enabledSelectors :=
      self perform: (self nextAbstractStateAfter: aMessage selector)
```

Example: Synchronization of a (Concurrent) Bounded Buffer (4)

```
EnabledSetsCountersActivity subclass: #BufferEnabledSetsCountersActivity
   instanceVariableNames: ''

"Abstract states for services availability constraints."
**As for enabled sets scheme, class BufferEnabledSetsActivity in Sect. 5.1**

"Guards (only) for intra-object concurrency synchronization."
guardOFget
   "Only one get at once."
   ^(self current: #get) = 0

guardOFput: item
   "Only one put: at once."
    ^(self current: #put:) = 0
```

With this mixed scheme, the clear separation of services availability and intra-object concurrency makes specifications more modular and consequently more reusable.

5.5 Further Extensions

The previous synchronization scheme (Sect. 5.4), although expressive, still cannot directly express constraints on method invocations such as "shortest job first served" policy. [MCH 94] proposed some modular and expressive synchronization scheme, named *Synchronization Variables*, which addresses such issues. We can quickly model and implement their scheme by augmenting previous synchronization counters model (1) by attaching specific information such as message arrival time or job priority onto invocations and (2) by providing various iteration and predicate methods over the mailbox (that is the ordered set of pending invocations). Such examples of refinements, which can express fine grain policies while ensuring liveness, have been implemented and described in [BRI 96]